
The Emacs Widget Library

Introduction

Most graphical user interface toolkits, such as Motif and XView, provide a number of standard
user interface controls (sometimes known as `widgets' or `gadgets'). Emacs doesn't really support
anything like this, except for an incredible powerful text \widget". On the other hand, Emacs
does provide the necessary primitives to implement many other widgets within a text bu�er.
The widget package simpli�es this task.

The basic widgets are:

link Areas of text with an associated action. Intended for hypertext links embedded in
text.

push-button
Like link, but intended for stand-alone buttons.

editable-field
An editable text �eld. It can be either variable or �xed length.

menu-choice
Allows the user to choose one of multiple options from a menu, each option is itself
a widget. Only the selected option will be visible in the bu�er.

radio-button-choice
Allows the user to choose one of multiple options by activating radio buttons. The
options are implemented as widgets. All options will be visible in the bu�er.

item A simple constant widget intended to be used in the menu-choice and radio-
button-choice widgets.

choice-item
A button item only intended for use in choices. When invoked, the user will be
asked to select another option from the choice widget.

toggle A simple `on'/`off' switch.

checkbox A checkbox (`[ ]'/`[X]').

editable-list
Create an editable list. The user can insert or delete items in the list. Each list
item is itself a widget.

Now of what possible use can support for widgets be in a text editor? I'm glad you asked.
The answer is that widgets are useful for implementing forms. A form in emacs is a bu�er where
the user is supposed to �ll out a number of �elds, each of which has a speci�c meaning. The
user is not supposed to change or delete any of the text between the �elds. Examples of forms
in Emacs are the `forms' package (of course), the customize bu�ers, the mail and news compose
modes, and the html form support in the `w3' browser.

The advantages for a programmer of using the widget package to implement forms are:
1. More complex �elds than just editable text are supported.
2. You can give the user immediate feedback if he enters invalid data in a text �eld, and

sometimes prevent entering invalid data.
3. You can have �xed sized �elds, thus allowing multiple �eld to be lined up in columns.
4. It is simple to query or set the value of a �eld.
5. Editing happens in bu�er, not in the mini-bu�er.

6. Packages using the library get a uniform look, making them easier for the user to learn.
7. As support for embedded graphics improve, the widget library will extended to support it.

This means that your code using the widget library will also use the new graphic features
by automatic.

In order to minimize the code that is loaded by users who does not create any widgets, the
code has been split in two �les:

`widget.el'
This will declare the user variables, de�ne the function define-widget, and au-
toload the function widget-create.

`wid-edit.el'
Everything else is here, there is no reason to load it explicitly, as it will be autoloaded
when needed.

User Interface

A form consists of read only text for documentation and some �elds, where each of the
�elds contains two parts, a tag and a value. The tags are used to identify the �elds, so the
documentation can refer to the foo �eld, meaning the �eld tagged with `Foo'. Here is an example
form:

Here is some documentation.

Name: My Name Choose: This option
Address: Some Place
In some City
Some country.

See also other work for more information.

Numbers: count to three below
[INS] [DEL] One
[INS] [DEL] Eh, two?
[INS] [DEL] Five!
[INS]

Select multiple:

[X] This
[ ] That
[X] Thus

Select one:

(*) One
( ) Another One.
( ) A Final One.

[Apply Form] [Reset Form]
The top level widgets in is example are tagged `Name', `Choose', `Address', `_other work_',

`Numbers', `Select multiple', `Select one', `[Apply Form]', and `[Reset Form]'. There are
basically two thing the user can do within a form, namely editing the editable text �elds and
activating the buttons.

Editable Text Fields

In the example, the value for the `Name' is most likely displayed in an editable text �eld,
and so are values for each of the members of the `Numbers' list. All the normal Emacs editing
operations are available for editing these �elds. The only restriction is that each change you
make must be contained within a single editable text �eld. For example, capitalizing all text
from the middle of one �eld to the middle of another �eld is prohibited.

Editing text �elds are created by the editable-field widget.
The editing text �elds are highlighted with the widget-field-face face, making them easy

to �nd.

Facewidget-�eld-face
Face used for other editing �elds.

Buttons

Some portions of the bu�er have an associated action, which can be invoked by a standard
key or mouse command. These portions are called buttons. The default commands for activating
a button are:

hRETi

Commandwidget-button-press pos &optional event
Invoke the button at pos, defaulting to point. If point is not located on
a button, invoke the binding in widget-global-map (by default the global
map).

mouse-2

Commandwidget-button-click event
Invoke the button at the location of the mouse pointer. If the mouse pointer
is located in an editable text �eld, invoke the binding in widget-global-map
(by default the global map).

There are several di�erent kind of buttons, all of which are present in the example:

The Option Field Tags.
When you invoke one of these buttons, you will be asked to choose between a number
of di�erent options. This is how you edit an option �eld. Option �elds are created
by the menu-choice widget. In the example, `Choose' is an option �eld tag.

The `[INS]' and `[DEL]' buttons.
Activating these will insert or delete elements from an editable list. The list is
created by the editable-list widget.

Embedded Buttons.
The ` other work ' is an example of an embedded button. Embedded buttons are not
associated with a �elds, but can serve any purpose, such as implementing hypertext
references. They are usually created by the link widget.

The `[ ]' and `[X] ' buttons.
Activating one of these will convert it to the other. This is useful for implementing
multiple-choice �elds. You can create it wit

The `( ) ' and `(*) ' buttons.
Only one radio button in a radio-button-choice widget can be selected at any
time. When you invoke one of the unselected radio buttons, it will be selected and
the previous selected radio button will become unselected.

The `[Apply Form] ' `[Reset Form]' buttons.
These are explicit buttons made with the push-button widget. The main di�erence
from the link widget is that the buttons are will be displayed as GUI buttons when
possible. enough.

To make them easier to locate, buttons are emphasized in the bu�er.

Facewidget-button-face
Face used for buttons.

User Optionwidget-mouse-face
Face used for buttons when the mouse pointer is above it.

Navigation

You can use all the normal Emacs commands to move around in a form bu�er, plus you will
have these additional commands:

hTABi

Commandwidget-forward &optional count
Move point count buttons or editing �elds forward.

hM-TABi

Commandwidget-backward &optional count
Move point count buttons or editing �elds backward.

Programming Example

Here is the code to implement the user interface example (see [User Interface], page 2).
(require 'widget)

(eval-when-compile
(require 'wid-edit))

(defvar widget-example-repeat)

(defun widget-example ()
"Create the widgets from the Widget manual."
(interactive)
(kill-buffer (get-buffer-create "*Widget Example*"))
(switch-to-buffer (get-buffer-create "*Widget Example*"))
(kill-all-local-variables)
(make-local-variable 'widget-example-repeat)
(widget-insert "Here is some documentation.\n\nName: ")
(widget-create 'editable-field

:size 13
"My Name")

(widget-create 'menu-choice
:tag "Choose"
:value "This"
:help-echo "Choose me, please!"



The Emacs Widget Library 5

:notify (lambda (widget &rest ignore)
(message "%s is a good choice!"

(widget-value widget)))
'(item :tag "This option" :value "This")
'(choice-item "That option")
'(editable-field :menu-tag "No option" "Thus option"))

(widget-insert "Address: ")
(widget-create 'editable-field

"Some Place\nIn some City\nSome country.")
(widget-insert "\nSee also ")
(widget-create 'link

:notify (lambda (&rest ignore)
(widget-value-set widget-example-repeat

'("En" "To" "Tre"))
(widget-setup))

"other work")
(widget-insert " for more information.\n\nNumbers: count to three below\n")
(setq widget-example-repeat

(widget-create 'editable-list
:entry-format "%i %d %v"
:notify (lambda (widget &rest ignore)

(let ((old (widget-get widget
':example-length))

(new (length (widget-value widget))))
(unless (eq old new)
(widget-put widget ':example-length new)
(message "You can count to %d." new))))

:value '("One" "Eh, two?" "Five!")
'(editable-field :value "three")))

(widget-insert "\n\nSelect multiple:\n\n")
(widget-create 'checkbox t)
(widget-insert " This\n")
(widget-create 'checkbox nil)
(widget-insert " That\n")
(widget-create 'checkbox

:notify (lambda (&rest ignore) (message "Tickle"))
t)

(widget-insert " Thus\n\nSelect one:\n\n")
(widget-create 'radio-button-choice

:value "One"
:notify (lambda (widget &rest ignore)

(message "You selected %s"
(widget-value widget)))

'(item "One") '(item "Another One.") '(item "A Final One."))
(widget-insert "\n")
(widget-create 'push-button

:notify (lambda (&rest ignore)
(if (= (length (widget-value widget-example-repeat))

3)
(message "Congratulation!")

(error "Three was the count!")))

"Apply Form")
(widget-insert " ")
(widget-create 'push-button

:notify (lambda (&rest ignore)
(widget-example))

"Reset Form")
(widget-insert "\n")
(use-local-map widget-keymap)
(widget-setup))

Setting Up the Bu�er

Widgets are created with widget-create, which returns a widget object. This object can
be queried and manipulated by other widget functions, until it is deleted with widget-delete.
After the widgets have been created, widget-setup must be called to enable them.

Functionwidget-create type [ keyword argument ]. . .
Create and return a widget of type type. The syntax for the type argument is described
in [Basic Types], page 7.

The keyword arguments can be used to overwrite the keyword arguments that are part of
type.

Functionwidget-delete widget
Delete widget and remove it from the bu�er.

Functionwidget-setup
Setup a bu�er to support widgets.

This should be called after creating all the widgets and before allowing the user to edit
them.

If you want to insert text outside the widgets in the form, the recommended way to do that
is with widget-insert.

Functionwidget-insert
Insert the arguments, either strings or characters, at point. The inserted text will be read
only.

There is a standard widget keymap which you might �nd useful.

Constwidget-keymap
A keymap with the global keymap as its parent.
hTABi and C-hTABi are bound to widget-forward and widget-backward, respectively.
hRETi and mouse-2are bound to widget-button-press and widget-button-.

Variablewidget-global-map
Keymap used by widget-button-press and widget-button-click when not on a button.
By default this is global-map.

Basic Types

The syntax of a type speci�cation is given below:
NAME ::= (NAME [KEYWORD ARGUMENT]... ARGS)

| NAME

where name is a widget name, keyword is the name of a property, argument is the value of
the property, and args are interpreted in a widget speci�c way.

There following keyword arguments that apply to all widgets:

:value The initial value for widgets of this type.

:format This string will be inserted in the bu�er when you create a widget. The following
`%' escapes are available:

`%['
`%]' The text inside will be marked as a button.

By default, the text will be shown in widget-button-face, and sur-
rounded by brackets.

User Optionwidget-button-pre�x
String to pre�x buttons.

User Optionwidget-button-su�x
String to su�x buttons.

`%{'
`%}' The text inside will be displayed in the face speci�ed by :sample-face.

`%v' This will be replaced with the bu�er representation of the widget's value.
What this is depends on the widget type.

`%d' Insert the string speci�ed by :doc here.

`%h' Like `%d', with the following modi�cations: If the documentation string
is more than one line, it will add a button which will toggle between
showing only the �rst line, and showing the full text. Furthermore, if
there is no :doc property in the widget, it will instead examine the
:documentation-property property. If it is a lambda expression, it
will be called with the widget's value as an argument, and the result
will be used as the documentation text.

`%t' Insert the string speci�ed by :tag here, or the princ representation of
the value if there is no tag.

`%%' Insert a literal `%'.

:button-face
Face used to highlight text inside %[ %] in the format.

:button-prefix
:button-suffix

Text around %[ %] in the format.
These can be

nil No text is inserted.

a string The string is inserted literally.

a symbol The value of the symbol is expanded according to this table.

:doc The string inserted by the `%d' or `%h' escape in the format string.

:tag The string inserted by the `%t' escape in the format string.

:tag-glyph
Name of image to use instead of the string speci�ed by `:tag' on Emacsen that
supports it.

:help-echo
Message displayed whenever you move to the widget with either widget-forward
or widget-backward.

:indent An integer indicating the absolute number of spaces to indent children of this widget.

:offset An integer indicating how many extra spaces to add to the widget's grandchildren
compared to this widget.

:extra-offset
An integer indicating how many extra spaces to add to the widget's children com-
pared to this widget.

:notify A function called each time the widget or a nested widget is changed. The function
is called with two or three arguments. The �rst argument is the widget itself, the
second argument is the widget that was changed, and the third argument is the
event leading to the change, if any. In editable �elds, this includes all insertions,
deletions, etc. To watch only for \�nal" actions, rede�ne the :action callback.

:menu-tag
Tag used in the menu when the widget is used as an option in a menu-choice widget.

:menu-tag-get
Function used for �nding the tag when the widget is used as an option in a menu-
choice widget. By default, the tag used will be either the :menu-tag or :tag
property if present, or the princ representation of the :value property if not.

:match Should be a function called with two arguments, the widget and a value, and re-
turning non-nil if the widget can represent the speci�ed value.

:validate
A function which takes a widget as an argument, and returns nil if the widget's cur-
rent value is valid for the widget. Otherwise it should return the widget containing
the invalid data, and set that widget's :error property to a string explaining the
error.
The following prede�ned function can be used:

Functionwidget-children-validate widget
All the :children of widget must be valid.

:tab-order
Specify the order in which widgets are traversed with widget-forward or widget-
backward. This is only partially implemented.
a. Widgets with tabbing order -1 are ignored.
b. (Unimplemented) When on a widget with tabbing order n, go to the next widget

in the bu�er with tabbing order n+1 or nil, whichever comes �rst.
c. When on a widget with no tabbing order speci�ed, go to the next widget in the

bu�er with a positive tabbing order, or nil

:parent The parent of a nested widget (e.g. a menu-choice item or an element of an
editable-list widget).

:sibling-args
This keyword is only used for members of a radio-button-choice or checklist.
The value should be a list of extra keyword arguments, which will be used when
creating the radio-button or checkbox associated with this item.

User Optionwidget-glyph-directory
Directory where glyphs are found. Widget will look here for a �le with the same name as
speci�ed for the image, with either a `.xpm' (if supported) or `.xbm' extension.

User Optionwidget-glyph-enable
If non-nil, allow glyphs to appear on displays where they are supported.

The link Widget

Syntax:
TYPE ::= (link [KEYWORD ARGUMENT]... [ VALUE ])

The value, if present, is used to initialize the :value property. The value should be a string,
which will be inserted in the bu�er.

By default the link will be shown in brackets.

User Optionwidget-link-pre�x
String to pre�x links.

User Optionwidget-link-su�x
String to su�x links.

The url-link Widget

Syntax:
TYPE ::= (url-link [KEYWORD ARGUMENT]... URL)

When this link is invoked, the www browser speci�ed by browse-url-browser-function
will be called with url .

The info-link Widget

Syntax:
TYPE ::= (info-link [KEYWORD ARGUMENT]... ADDRESS)

When this link is invoked, the built-in info browser is started on address.

The push-button Widget

Syntax:
TYPE ::= (push-button [KEYWORD ARGUMENT]... [ VALUE ])

The value, if present, is used to initialize the :value property. The value should be a string,
which will be inserted in the bu�er.

By default the tag will be shown in brackets.

User Optionwidget-push-button-pre�x
String to pre�x push buttons.

User Optionwidget-push-button-su�x
String to su�x push buttons.

The editable-field Widget

Syntax:

TYPE ::= (editable-field [KEYWORD ARGUMENT]... [ VALUE ])

The value, if present, is used to initialize the :value property. The value should be a string,
which will be inserted in �eld. This widget will match all string values.

The following extra properties are recognized.

:size The minimum width of the editable �eld.
By default the �eld will reach to the end of the line. If the content is too large, the
displayed representation will expand to contain it. The content is not truncated to
size.

:value-face
Face used for highlighting the editable �eld. Default is widget-field-face.

:secret Character used to display the value. You can set this to e.g. ?* if the �eld contains
a password or other secret information. By default, the value is not secret.

:valid-regexp
By default the :validate function will match the content of the �eld with the value
of this attribute. The default value is "" which matches everything.

:keymap Keymap used in the editable �eld. The default value is widget-field-keymap,
which allows you to use all the normal editing commands, even if the bu�er's major
mode suppress some of them. Pressing return invokes the function speci�ed by
:action.

The text Widget

This is just like editable-field, but intended for multiline text �elds. The default :keymap
is widget-text-keymap, which does not rebind the return key.

The menu-choice Widget

Syntax:

TYPE ::= (menu-choice [KEYWORD ARGUMENT]... TYPE ... )

The type argument represents each possible choice. The widget's value will be that of the
chosen type argument. This widget will match any value matching at least one of the speci�ed
type arguments.

:void Widget type used as a fallback when the value does not match any of the speci�ed
type arguments.

:case-fold
Set this to nil if you don't want to ignore case when prompting for a choice through
the minibu�er.

:children
A list whose car is the widget representing the currently chosen type in the bu�er.

:choice The current chosen type

:args The list of types.

The radio-button-choice Widget

Syntax:
TYPE ::= (radio-button-choice [KEYWORD ARGUMENT]... TYPE ... )

The type argument represents each possible choice. The widget's value will be that of the
chosen type argument. This widget will match any value matching at least one of the speci�ed
type arguments.

The following extra properties are recognized.

:entry-format
This string will be inserted for each entry in the list. The following `%' escapes are
available:

`%v' Replaced with the bu�er representation of the type widget.

`%b' Replace with the radio button.

`%%' Insert a literal `%'.

button-args
A list of keywords to pass to the radio buttons. Useful for setting e.g. the
`:help-echo' for each button.

:buttons The widgets representing the radio buttons.

:children
The widgets representing each type.

:choice The current chosen type

:args The list of types.

You can add extra radio button items to a radio-button-choice widget after it has been
created with the function widget-radio-add-item.

Functionwidget-radio-add-item widget type
Add to radio-button-choice widget widget a new radio button item of type type.

Please note that such items added after the radio-button-choice widget has been created
will not be properly destructed when you call widget-delete.

The item Widget

Syntax:
ITEM ::= (item [KEYWORD ARGUMENT]... VALUE)

The value, if present, is used to initialize the :value property. The value should be a string,
which will be inserted in the bu�er. This widget will only match the speci�ed value.

The choice-item Widget

Syntax:
ITEM ::= (choice-item [KEYWORD ARGUMENT]... VALUE)

The value, if present, is used to initialize the :value property. The value should be a string,
which will be inserted in the bu�er as a button. Activating the button of a choice-item is
equivalent to activating the parent widget. This widget will only match the speci�ed value.

The toggle Widget

Syntax:
TYPE ::= (toggle [KEYWORD ARGUMENT]...)

The widget has two possible states, `on' and `o�', which correspond to a t or nil value
respectively.

The following extra properties are recognized.

:on String representing the `on' state. By default the string `on'.

:off String representing the `o�' state. By default the string `off'.

:on-glyph
Name of a glyph to be used instead of the `:on' text string, on emacsen that supports
it.

:off-glyph
Name of a glyph to be used instead of the `:o�' text string, on emacsen that supports
it.

The checkbox Widget

The widget has two possible states, `selected' and `unselected', which corresponds to a t or
nil value.

Syntax:
TYPE ::= (checkbox [KEYWORD ARGUMENT]...)

The checklist Widget

Syntax:
TYPE ::= (checklist [KEYWORD ARGUMENT]... TYPE ... )

The type arguments represents each checklist item. The widget's value will be a list con-
taining the values of all ticked type arguments. The checklist widget will match a list whose
elements all match at least one of the speci�ed type arguments.

The following extra properties are recognized.

:entry-format
This string will be inserted for each entry in the list. The following `%' escapes are
available:

`%v' Replaced with the bu�er representation of the type widget.

`%b' Replace with the checkbox.

`%%' Insert a literal `%'.

:greedy Usually a checklist will only match if the items are in the exact sequence given in
the speci�cation. By setting :greedy to non-nil, it will allow the items to appear in
any sequence. However, if you extract the values they will be in the sequence given
in the checklist. I.e. the original sequence is forgotten.

button-args
A list of keywords to pass to the checkboxes. Useful for setting e.g. the `:help-echo'
for each checkbox.

:buttons The widgets representing the checkboxes.

:children
The widgets representing each type.

:args The list of types.

The editable-list Widget

Syntax:
TYPE ::= (editable-list [KEYWORD ARGUMENT]... TYPE)

The value is a list, where each member represents one widget of type type.
The following extra properties are recognized.

:entry-format
This string will be inserted for each entry in the list. The following `%' escapes are
available:

`%v' This will be replaced with the bu�er representation of the type widget.

`%i' Insert the [INS] button.

`%d' Insert the [DEL] button.

`%%' Insert a literal `%'.

:insert-button-args
A list of keyword arguments to pass to the insert buttons.

:delete-button-args
A list of keyword arguments to pass to the delete buttons.

:append-button-args
A list of keyword arguments to pass to the trailing insert button.

:buttons The widgets representing the insert and delete buttons.

:children
The widgets representing the elements of the list.

:args List whose car is the type of the list elements.

The group Widget

This widget simply groups other widgets together.
Syntax:

TYPE ::= (group [KEYWORD ARGUMENT]... TYPE...)

The value is a list, with one member for each type.

Sexp Types

A number of widgets for editing s-expressions (lisp types) are also available. These basically
fall in the following categories.

The Constant Widgets.

The const widget can contain any lisp expression, but the user is prohibited from editing it,
which is mainly useful as a component of one of the composite widgets.

The syntax for the const widget is
TYPE ::= (const [KEYWORD ARGUMENT]... [ VALUE ])

The value, if present, is used to initialize the :value property and can be any s-expression.

Widgetconst
This will display any valid s-expression in an immutable part of the bu�er.

There are two variations of the const widget, namely variable-item and function-item.
These should contain a symbol with a variable or function binding. The major di�erence from
the const widget is that they will allow the user to see the variable or function documentation
for the symbol.

Widgetvariable-item
An immutable symbol that is bound as a variable.

Widgetfunction-item
An immutable symbol that is bound as a function.

Generic Sexp Widget.

The sexp widget can contain any lisp expression, and allows the user to edit it inline in the
bu�er.

The syntax for the sexp widget is
TYPE ::= (sexp [KEYWORD ARGUMENT]... [ VALUE ])

Widgetsexp
This will allow you to edit any valid s-expression in an editable bu�er �eld.
The sexp widget takes the same keyword arguments as the editable-field widget.

Atomic Sexp Widgets.

The atoms are s-expressions that does not consist of other s-expressions. A string is an atom,
while a list is a composite type. You can edit the value of an atom with the following widgets.

The syntax for all the atoms are
TYPE ::= (NAME [KEYWORD ARGUMENT]... [ VALUE ])

The value, if present, is used to initialize the :value property and must be an expression of
the same type as the widget. I.e. the string widget can only be initialized with a string.

All the atom widgets take the same keyword arguments as the editable-field widget.

Widgetstring
Allows you to edit a string in an editable �eld.

Widgetregexp
Allows you to edit a regular expression in an editable �eld.

Widgetcharacter
Allows you to enter a character in an editable �eld.

Widget�le
Allows you to edit a �le name in an editable �eld. If you invoke the tag button, you can
edit the �le name in the mini-bu�er with completion.
Keywords:

:must-match
If this is set to non-nil, only existing �le names will be allowed in the mini-
bu�er.

Widgetdirectory
Allows you to edit a directory name in an editable �eld. Similar to the file widget.

Widgetsymbol
Allows you to edit a lisp symbol in an editable �eld.

Widgetfunction
Allows you to edit a lambda expression, or a function name with completion.

Widgetvariable
Allows you to edit a variable name, with completion.

Widgetinteger
Allows you to edit an integer in an editable �eld.

Widgetnumber
Allows you to edit a number in an editable �eld.

Widgetboolean
Allows you to edit a boolean. In lisp this means a variable which is either nil meaning
false, or non-nil meaning true.

Composite Sexp Widgets.

The syntax for the composite are
TYPE ::= (NAME [KEYWORD ARGUMENT]... COMPONENT...)

Where each componentmust be a widget type. Each component widget will be displayed in
the bu�er, and be editable to the user.

Widgetcons
The value of a cons widget is a cons-cell where the car is the value of the �rst component
and the cdr is the value of the second component. There must be exactly two components.

Widgetlist
The value of a list widget is a list containing the value of each of its component.

Widgetvector
The value of a vector widget is a vector containing the value of each of its component.

The above su�ce for specifying �xed size lists and vectors. To get variable length lists and
vectors, you can use a choice, set or repeat widgets together with the :inline keywords. If
any component of a composite widget has the :inline keyword set, its value must be a list
which will then be spliced into the composite. For example, to specify a list whose �rst element
must be a �le name, and whose remaining arguments should either by the symbol t or two �les,
you can use the following widget speci�cation:

(list file
(choice (const t)

(list :inline t
:value ("foo" "bar")
string string)))

The value of a widget of this type will either have the form `(file t)' or (file string
string).

This concept of inline is probably hard to understand. It was certainly hard to implement
so instead of confusing you more by trying to explain it here, I'll just suggest you meditate over
it for a while.

Widgetchoice
Allows you to edit a sexp which may have one of a �xed set of types. It is currently
implemented with the choice-menu basic widget, and has a similar syntax.

Widgetset
Allows you to specify a type which must be a list whose elements all belong to given set.
The elements of the list is not signi�cant. This is implemented on top of the checklist
basic widget, and has a similar syntax.

Widgetrepeat
Allows you to specify a variable length list whose members are all of the same type.
Implemented on top of the `editable-list' basic widget, and has a similar syntax.

Properties

You can examine or set the value of a widget by using the widget object that was returned
by widget-create.

Functionwidget-value widget
Return the current value contained in widget. It is an error to call this function on an
uninitialized widget.

Functionwidget-value-set widget value
Set the value contained in widget to value. It is an error to call this function with an
invalid value.

Important: You must call widget-setup after modifying the value of a widget before the
user is allowed to edit the widget again. It is enough to call widget-setup once if you modify
multiple widgets. This is currently only necessary if the widget contains an editing �eld, but
may be necessary for other widgets in the future.

If your application needs to associate some information with the widget objects, for example
a reference to the item being edited, it can be done with widget-put and widget-get. The
property names must begin with a `:'.

Functionwidget-put widget property value
In widget set property to value. property should be a symbol, while value can be anything.

Functionwidget-get widget property
In widget return the value for property . property should be a symbol, the value is what
was last set by widget-put for property .

Functionwidget-member widget property
Non-nil if widget has a value (even nil) for property property .

Occasionally it can be useful to know which kind of widget you have, i.e. the name of the
widget type you gave when the widget was created.

Functionwidget-type widget
Return the name of widget, a symbol.

Widgets can be in two states: active, which means they are modi�able by the user, or
inactive, which means they cannot be modi�ed by the user. You can query or set the state with
the following code:

;; Examine if widget is active or not.
(if (widget-apply widget :active)

(message "Widget is active.")
(message "Widget is inactive.")

;; Make widget inactive.
(widget-apply widget :deactivate)

;; Make widget active.
(widget-apply widget :activate)

A widget is inactive if itself or any of its ancestors (found by following the :parent link)
have been deactivated. To make sure a widget is really active, you must therefore activate both
itself and all its ancestors.

(while widget
(widget-apply widget :activate)
(setq widget (widget-get widget :parent)))

You can check if a widget has been made inactive by examining the value of the :inactive
keyword. If this is non-nil, the widget itself has been deactivated. This is di�erent from using
the :active keyword, in that the latter tells you if the widget or any of its ancestors have
been deactivated. Do not attempt to set the :inactive keyword directly. Use the :activate
:deactivate keywords instead.

De�ning New Widgets

You can de�ne specialized widgets with define-widget. It allows you to create a shorthand
for more complex widgets. This includes specifying component widgets and new default values
for the keyword arguments.

Functionde�ne-widget name class doc&rest args
De�ne a new widget type named name from class.
name and class should both be symbols, class should be one of the existing widget types.
The third argument DOC is a documentation string for the widget.
After the new widget has been de�ned the following two calls will create identical widgets:
�

(widget-create name)
�

(apply widget-create class args)

Using define-widget just stores the de�nition of the widget type in the widget-type prop-
erty of name, which is what widget-create uses.

If you just want to specify defaults for keywords with no complex conversions, you can use
identity as your :convert-widget function.

The following additional keyword arguments are useful when de�ning new widgets:

:convert-widget
Method to convert type-speci�c components of a widget type before instantiating a
widget of that type. Not normally called from user code, it is invoked by widget-
convert. Typical operations include converting types of child widgets to widget

instances and converting values from external format (i.e., as expected by the calling
code) to internal format (which is often di�erent for the convenience of widget
manipulation). It takes a widget type as an argument, and returns the converted
widget type. When a widget is created, the value of this property is called for the
widget type, then for all the widget's parent types, most derived �rst. (The property
is reevaluated for each parent type.)
The following prede�ned functions can be used here:

Functionwidget-types-convert-widget widget
Convert each member of :args in widget from a widget type to a widget.

Functionwidget-value-convert-widget widget
Initialize :value from (car :args) in widget, and reset :args.

:copy A method to implement deep copying of the type. Any member of the widget which
might be changed in place (rather than replaced) should be copied by this method.
(widget-copy uses copy-sequence to ensure that the top-level list is a copy.) This
particularly applies to child widgets.

:value-to-internal
Function to convert the value to the internal format. The function takes two argu-
ments, a widget and an external value. It returns the internal value. The function
is called on the present :value when the widget is created, and on any value set
later with widget-value-set.

:value-to-external
Function to convert the value to the external format. The function takes two argu-
ments, a widget and an internal value, and returns the internal value. The function
is called on the present :value when the widget is created, and on any value set
later with widget-value-set.

:create Function to create a widget from scratch. The function takes one argument, a
widget, and inserts it in the bu�er. Not normally called from user code. Instead,
call widget-create or related functions, which take a type argument, (usually)
convert it to a widget, call the :create function to insert it in the bu�er, and then
return the (possibly converted) widget.
The default, widget-default-create, is invariably appropriate. (None of the stan-
dard widgets specify :create.)

:delete Function to delete a widget. The function takes one argument, a widget, and should
remove all traces of the widget from the bu�er.

:value-create
Function to expand the `%v' escape in the format string. It will be called with the
widget as its argument and should insert a representation of the widget's value in
the bu�er.

:value-delete
Should remove the representation of the widget's value from the bu�er. It will be
called with the widget as its argument. It doesn't have to remove the text, but it
should release markers and delete nested widgets if such have been used.
The following prede�ned function can be used here:

Functionwidget-children-value-delete widget
Delete all :children and :buttons in widget.

:value-get
Function to extract the value of a widget, as it is displayed in the bu�er.
The following prede�ned function can be used here:

Functionwidget-value-value-get widget
Return the :value property of widget.

:format-handler
Function to handle unknown `%' escapes in the format string. It will be called with
the widget and the escape character as arguments. You can set this to allow your
widget to handle non-standard escapes.
You should end up calling widget-default-format-handler to handle unknown
escape sequences. It will handle the `%h' and any future escape sequences as well as
give an error for unknown escapes.

:action Function to handle user initiated events. By default, :notify the parent. Actions
normally do not include mere edits, but refer to things like invoking buttons or
hitting enter in an editable �eld. To watch for any change, rede�ne the :notify
callback.
The following prede�ned function can be used here:

Functionwidget-parent-action widget &optional event
Tell :parent of widget to handle the :action.
Optional event is the event that triggered the action.

:prompt-value
Function to prompt for a value in the minibu�er. The function should take four
arguments, widget, prompt , value, and unbound and should return a value for widget
entered by the user. prompt is the prompt to use. value is the default value to use,
unless unbound is non-nil. In this case there is no default value. The function
should read the value using the method most natural for this widget and does not
have to check whether it matches.

If you want to de�ne a new widget from scratch, use the default widget as its base.

Widgetdefault
Widget used as a base for other widgets.
It provides most of the functionality that is referred to as \by default" in this text.

In implementing complex hierarchical widgets (e.g., using the `group' widget), the following
functions may be useful. The syntax for the type arguments to these functions is described in
[Basic Types], page 7.

Functionwidget-create-child-and-convert parent type &rest args
As a child of parent, create a widget with type type and value value. type is copied, and
the :widget-contvert method is applied to the optional keyword arguments from args.

Functionwidget-create-child parent type
As a child of parent, create a widget with type type. type is copied, but no conversion
method is applied.

Functionwidget-create-child-value parent type value
As a child of parent, create a widget with type type and value value. type is copied, but
no conversion method is applied.

Functionwidget-convert type &rest args
Convert type to a widget without inserting it in the bu�er. The optional args are addi-
tional keyword arguments.
The widget's :args property is set from the longest tail of args whose `cdr' is not a
keyword, or if that is null, from the longest tail of type's :args property whose cdr is not
a keyword. Keyword arguments from args are set, and the :value property (if any) is
converted from external to internal format.

widget-convert is typically not called from user code; rather it is called implicitly through
the `widget-create*' functions.

Widget Browser

There is a separate package to browse widgets. This is intended to help programmers who
want to examine the content of a widget. The browser shows the value of each keyword, but
uses links for certain keywords such as `:parent', which avoids printing cyclic structures.

Commandwidget-browse WIDGET
Create a widget browser for WIDGET. When called interactively, prompt for WIDGET.

Commandwidget-browse-other-window WIDGET
Create a widget browser for WIDGET and show it in another window. When called
interactively, prompt for WIDGET.

Commandwidget-browse-at POS
Create a widget browser for the widget at POS. When called interactively, use the position
of point.

Widget Minor Mode

There is a minor mode for manipulating widgets in major modes that doesn't provide any
support for widgets themselves. This is mostly intended to be useful for programmers doing
experiments.

Commandwidget-minor-mode
Toggle minor mode for traversing widgets. With arg, turn widget mode on if and only if
arg is positive.

Variablewidget-minor-mode-keymap
Keymap used in widget-minor-mode.

Utilities.

Functionwidget-prompt-value widget prompt [ value unbound ]
Prompt for a value matching widget, using prompt .
The current value is assumed to be value, unless unbound is non-nil.

Functionwidget-get-sibling widget
Get the item widget is assumed to toggle.
This is only meaningful for radio buttons or checkboxes in a list.

Wishlist

� It should be possible to add or remove items from a list with C-k and C-o (suggested by
rms ).

� The `[INS]' and `[DEL]' buttons should be replaced by a single dash (`-'). The dash should
be a button that, when invoked, ask whether you want to add or delete an item (rms wanted
to git rid of the ugly buttons, the dash is my idea).

� The menu-choice tag should be prettier, something like the abbreviated menus in Open
Look.

� Finish :tab-order.
� Make indentation work with glyphs and proportional fonts.
� Add commands to show overview of object and class hierarchies to the browser.
� Find a way to disable mouse highlight for inactive widgets.
� Find a way to make glyphs look inactive.
� Add property-list widget.
� Add association-list widget.
� Add key-binding widget.
� Add widget widget for editing widget speci�cations.
� Find clean way to implement variable length list. See TeX-printer-list for an explanation.
� C-h in widget-prompt-value should give type speci�c help.
� A mailto widget.
� C-e e in a �xed size �eld should go to the end of the text in the �eld, not the end of the

�eld itself.
� Use an overlay instead of markers to delimit the widget. Create accessors for the end points.
� Clicking on documentation links should call describe-function or widget-browse-other-

window and friends directly, instead of going through apropos. If more than one function
is valid for the symbol, it should pop up a menu.

Internals

This (very brief!) section provides a few notes on the internal structure and implementation
of Emacs widgets. Avoid relying on this information. (We intend to improve it, but this will
take some time.) To the extent that it actually describes APIs, the information will be moved
to appropriate sections of the manual in due course.

The Widget and Type Structures

Widgets and types are currently both implemented as lists.
A symbol may be de�ned as a type name using define-widget. See [De�ning New Widgets],

page 17. A type is a list whose car is a previously de�ned type name, nil, or (recursively) a
type. The car is the classor parent type of the type, and properties which are not speci�ed in
the new type will be inherited from ancestors. Probably the only type without a class should
be the default type. The cdr of a type is a plist whose keys are widget property keywords.

A type or type name may also be referred to as an unconverted widget.
A converted widget or widget instance is a list whose car is a type name or a type, and

whose cdr is a property list. Furthermore, all children of the converted widget must be converted.
Finally, in the process of appropriate parts of the list structure are copied to ensure that changes
in values of one instance do not a�ect another's.

